
1 [bookmark: _Toc122434485][bookmark: _Toc97884702][bookmark: _Toc97906423] Overview
This document lists all the essencial changes needed to correct problems in the TTCN implementation of NR5GC test case 8.1.4.4.4 which is part of the NR5GC test suite in iWD_TTCN3-B2020-09_D21wk49.
With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log file(s) is (are) provided as evidence. 
Contact:	Parikshit Bhise
	parikshit.bhise@rohde-schwarz.com 
	Narendra Kalahasti
	narendra.kalahasti@anritsu.com
	
1. [bookmark: _Toc97884703][bookmark: _Toc97906424]Verification Test Summary 
Test Case: 	8.1.4.4.4.NR5GC
ATS Version:	iwd-TTCN3-B2020-09_D21wk49
[bookmark: _Hlk37919671]System Simulator used:	R&S® 5G Protocol Conformance Test platform
UE used:	MTK 6983 UE
Verification Status:	PASS
2 [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc97884704][bookmark: _Toc97906425]Corrections required to test case 8.1.4.4.4
2.1 [bookmark: _Toc97128120][bookmark: _Toc97906426]Correction to the function fl_TC_8_1_4_4_4_TestBody ()
	Function name
	fl_TC_8_1_4_4_4_TestBody()

	Reason for change
	1. As per prose R5-220605 IP data path verification was missing at step 4A and 9A.
2. After Step 12 UE detects handover failure on NR Cell 1 and in order to meet UE can not find any selectable cells after T304 timer expiry a T2 power criteria needs to be introduced. i.e., before T311 Power levels are such that all Srxlev NR Cell < 0 and after that T3 powel levels are such that NR Cell 4 is better cell meeting the event A3 criteria.
3. At step 3 MasterKeyUpdate Condition according to table Table 8.1.4.4.4.3.3-9 TS 38.523-1 with condition RBConfig_KeyChange is missing to enable.
4. Rach reception for T304 timer is required on NR Cell 1as per Step7 and Step 12 along with Rach indication enabling and disabling.
5. After Step 14 the UE wil be indle state, so it cannot transmit rrcReconfigurationComplete. In order to ensure UE is in idle state it is proposed that a mobilty registration procedure on NR Cell 4 is performed instead. A Prose CR will be raised for this. At Step 13A to ensure T304 and T311 expire 10s timer is bit longer.
6. After Step 13A when Srxlev NR Cell < 0 the UE is in idle state. However, the local end is not updated correspondingly. In order to perform a mobility Registration on NR Cell 4, it is proposed to release and reconfigure the NR Cell 4.


	Summary of change
	1. Added data path flow testing at step 4A and 9A.
2. Updated T2 powel level so that Srxlev NR Cell < 0 for all NR Cells and added T3 power level to meet A3 event for NR Cell 4 only.
3. At Step 3 added masterKeyUpdate template while calling f_NR5GC_508RRC_IntraNR_HO_InterCell function.
4. T304 timer requirement for Step7 and Step 12 Rach indication implemented and handled via newly added altstep under activate/decactivate mechanism.
5. Added mobility registration updating procedure after Step 14 just to ensure UE is in idle state and perfrom mobility registration on NR Cell 4 after T3 powel level.
6. At Step 13A to ensure T304 and T311 expire 5s timer is enough.
7. Called function f_NR_RRCRelease_Local for NR Cell4.

	TTCN module
	RRC_Conditional_Handover_NR5GC.ttcn

	MCC160 Comment
	1. Accepted, already implemented in wk12
2. Accepted, the associated prose CR R5-223358 has been agreed in RAN5#95 e-meeting.
3. Accepted
4. [bookmark: _GoBack]Accepted
5. Accepted, the associated prose CR R5-223358 has been agreed in RAN5#95 e-meeting.
6. Accepted, the associated prose CR R5-223358 has been agreed in RAN5#95 e-meeting.
7. Accepted





Before Change:
	  function fl_TC_8_1_4_4_4_TestBody() runs on NR5GC_PTC
  {
    var template (value) NR_CellPowerList_Type v_CellPowerList_AtT1;
    var template (value) NR_CellPowerList_Type v_CellPowerList_AtT2;
    var NR_AbsoluteCellPower_Type v_CellPower_Serving_FR1;
    var NR_AbsoluteCellPower_Type v_CellPower_EntryCondition_FR1;
    var NR_AbsoluteCellPower_Type v_CellPower_Leaving_FR1;
    var RSRP_Range v_Rsrp;
    var template (value) MeasConfig v_MeasConfig;
    var template (value) ConditionalReconfiguration_r16 v_ConditionalReconfiguration_r16_Step;
    //var template (value) ConditionalReconfiguration_r16 v_ConditionalReconfiguration_r16_Step_5;
    var template (value) ConditionalReconfiguration_r16 v_ConditionalReconfiguration_r16_Step_10;
    var template (value) NR_DRBInfoList_Type v_DRBInfoList := f_NR5GC_MobileInfo_GetDRBInfoList();
    var template (value) DL_DCCH_Message v_RRCReconfigurationHO;
    var float v_T304Min := f_NR_SetTimerToleranceMin(nr_Cell1, rrcTimer, 1.0); //T304 is ms1000 according to 38.508-1 Table 4.6.3-19.
    var float v_T311Min := f_NR_SetTimerToleranceMin(nr_Cell1, rrcTimer, 1.0); //T311 is ms1000 according to 38.508-1 Table 4.6.3-150.
    var float v_T304_T311Sum := v_T304Min + v_T311Min;
    timer t_Watchdog := 10.0;
    timer t_T304Min;

    //Power levels of NR cell 1, NR cell 2 and NR cell 4 FR1 acc. Table 8.1.4.4.1.3.2-1, Table 8.1.4.4.1.3.2-2 FFS
    /*         |   Cell 1   |  Cell 2  |  Cell 4
     * -----------------------------------------
     *   T0    |    -85     |   -113   |   -88
     * -----------------------------------------
     *   T1    |    -85     |   -79    |   -113
     * -----------------------------------------
     *   T2    |    -85     |   -113   |   -79
     */

    v_CellPower_Serving_FR1 := -85;
    v_CellPower_Leaving_FR1 := -113;
    v_CellPower_EntryCondition_FR1 := -79;

    v_CellPowerList_AtT1 := {cs_NR_CellPower(nr_Cell1, v_CellPower_Serving_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE),  // FR2 is FFS
                             cs_NR_CellPower(nr_Cell2, v_CellPower_EntryCondition_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE), // FR2 is FFS
                             cs_NR_CellPower(nr_Cell4, v_CellPower_Leaving_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE)}; // FR2 is FFS
    v_CellPowerList_AtT2 := {cs_NR_CellPower(nr_Cell1, v_CellPower_Serving_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE),  // FR2 is FFS
                             cs_NR_CellPower(nr_Cell2, v_CellPower_Leaving_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE),  // FR2 is FFS
                             cs_NR_CellPower(nr_Cell4, v_CellPower_EntryCondition_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE)}; // FR2 is FFS

    if(f_NR_CellInfo_GetIsFR1(nr_Cell1)){
       v_Rsrp := 2;
    }
    else{
       v_Rsrp := 2;          //FFS dummy value FR2
    }

    // Reconfigure nr_Cell2 to no respond to any RA Preamble
    // Table 8.1.4.4.4.3.2-4: The UE attempts to perform the inter frequency handover using MAC Random Access Preamble on NR Cell 1. SS does not respond.
    f_NR_SS_RachProcedureConfig(nr_Cell1, cs_NR_RachProcedureConfig_NoResponse);

    //@siclog "Steps 1-2" siclog@
    //The SS transmits an RRCReconfiguration message including MeasConfig and ConditionalReconfiguration to set NR Cell 2 as the target candidate cell.
    //The UE transmits an RRCReconfigurationComplete message in NR cell 1.
    v_MeasConfig := f_NR_GenerateMeasurementConfigNR( {cs_MeasObjectId1(nr_Cell1)},
                                                      {cs_NR_ReportConfigCondTrigger(tsc_NR_IdReportConfigId1, v_Rsrp)},
                                                      {cs_NR_MeasId_Config_id1_obj1_conf1}
                                                      );
    v_ConditionalReconfiguration_r16_Step := cs_ConditionalReconfiguration_r16( -,
                                                                                cs_CondReconfigToAddModList_r16_1EntryCR_1EntryMI(1,
                                                                                                                                  tsc_NR_MeasId1,
                                                                                                                                  f_NR_GenerateCondRRCReconfig_r16_Octetstring(nr_Cell2)));
    SRB.send(cas_NR_SRB1_RrcPdu_REQ( nr_Cell1,
                                     cs_TimingInfo_Now,
                                     cs_38508_RRCReconfiguration_CHO(-,v_MeasConfig, v_ConditionalReconfiguration_r16_Step)));
    SRB.receive(car_NR_SRB1_RrcPdu_IND(nr_Cell1, cr_38508_RRCReconfigurationComplete));

    //@siclog "Steps 3-4" siclog@
    //The SS transmits an RRCReconfiguration message including a reconfigurationWithSync for handover to NR Cell 4.
    //Check: Does UE transmit the RRCReconfigurationComplete message in NR Cell 4 within 10s?
    f_NR5GC_508RRC_IntraNR_HO_InterCell(nr_Cell1,
                                        nr_Cell4,
                                        -,
                                        - // RBConfig_KeyChange := true
                                        );
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 4");

    //@siclog "Steps 5-6" siclog@
    //The SS transmits an RRCReconfiguration message including MeasConfig and ConditionalReconfiguration to set NR Cell 2 as the target candidate cell.
    //The UE transmits an RRCReconfigurationComplete message in NR Cell 4.
    //v_ConditionalReconfiguration_r16_Step_5 := cs_ConditionalReconfiguration_r16( -,
                                                                                 // cs_CondReconfigToAddModList_r16_1EntryCR_1EntryMI(1,
                                                                                                                                    //tsc_NR_MeasId1,
                                                                                                                                    //f_NR_GenerateCondRRCReconfig_r16_Octetstring(nr_Cell2)));
    f_NR5GC_CHO_RRCReconfiguration_SSConfig( nr_Cell4,
                                             nr_Cell2,
                                             v_DRBInfoList,
                                             -,-,
                                             cs_38508_RRCReconfiguration_CHO(-,v_MeasConfig, v_ConditionalReconfiguration_r16_Step)
                                             );

    //@siclog "Steps 7" siclog@
    //The SS transmits an RRCReconfiguration message including a reconfigurationWithSync for handover to NR Cell 1.
    v_RRCReconfigurationHO := f_NR5GC_RRCReconfiguration_IntraNR_HO(nr_Cell1);
    SRB.send(cas_NR_SRB1_RrcPdu_REQ(nr_Cell4, cs_TimingInfo_Now, v_RRCReconfigurationHO));
    t_T304Min.start(v_T304Min);

    //EXCEPTION: In parallel to the events described in step 10 the steps specified in Table 8.1.4.4.4.3.2-4 should take place.
    //Nothing to do nr_Cell1 already configured to not respond to any RA Preamble

    //@siclog "Step 8" siclog@
    //SS adjusts the cell-specific reference signal level according to row "T1".
    f_NR_SetCellPowerList(v_CellPowerList_AtT1);

    //@siclog "Step 9" siclog@
    //Check: Does UE transmit the RRCReconfigurationComplete message in NR Cell 2 within 10s?
    t_Watchdog.start;
    f_NR5GC_CHO_RRCReconfigurationComplete( nr_Cell4,
                                            nr_Cell2,
                                            -);
    t_Watchdog.stop;
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 9");

    //@siclog "Steps 10-11" siclog@
    //The SS transmits an RRCReconfiguration message including MeasConfig and ConditionalReconfiguration to set NR Cell 4 as the target candidate cell.
    //The UE transmits an RRCReconfigurationComplete message in NR Cell 2.
    v_ConditionalReconfiguration_r16_Step_10 := cs_ConditionalReconfiguration_r16( -,
                                                                                  cs_CondReconfigToAddModList_r16_1EntryCR_1EntryMI(1,
                                                                                                                                    tsc_NR_MeasId1,
                                                                                                                                    f_NR_GenerateCondRRCReconfig_r16_Octetstring(nr_Cell4)));
    SRB.send(cas_NR_SRB1_RrcPdu_REQ( nr_Cell2,
                                     cs_TimingInfo_Now,
                                     cs_38508_RRCReconfiguration_CHO(-,v_MeasConfig, v_ConditionalReconfiguration_r16_Step_10)));
    SRB.receive(car_NR_SRB1_RrcPdu_IND(nr_Cell2, cr_38508_RRCReconfigurationComplete));

    //@siclog "Steps 12" siclog@
    //The SS transmits an RRCReconfiguration message including a reconfigurationWithSync for handover to NR Cell 1.
    v_RRCReconfigurationHO := f_NR5GC_RRCReconfiguration_IntraNR_HO(nr_Cell1);
    SRB.send(cas_NR_SRB1_RrcPdu_REQ(nr_Cell2, cs_TimingInfo_Now, v_RRCReconfigurationHO));

    //EXCEPTION: In parallel to the events described in step 16 the steps specified in Table 8.1.4.4.4.3.2-4 should take place.
    //Nothing to do nr_Cell1 already configured to not respond to any RA Preamble

    //@siclog "Step 13" siclog@
    //SS wait 10s to make sure T304+T311 expire. (Note 1)
    //Note 1: The wait time is selected to cover T304 (1000ms) + T311 (1000ms) to ensure that UE goes to RRC_IDLE.
    f_Delay(v_T304_T311Sum);

    //@siclog "Step 14" siclog@
    //SS adjusts the cell-specific reference signal level according to row "T2".
    f_NR_SetCellPowerList(v_CellPowerList_AtT2);

    //@siclog "Step 15" siclog@
    //Check: Does UE transmit the RRCReconfigurationComplete message in NR Cell 2 within 10s?
    t_Watchdog.start;
    alt {
        []SRB.receive(car_NR_SRB1_RrcPdu_IND(nr_Cell2, cr_38508_RRCReconfigurationComplete))
        {
            f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 15, RRCReconfigurationComplete sent by UE ");
        }
        []t_Watchdog.timeout
        {
            f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 15");
        }
    };

  } // End of f_TC_8_1_4_4_4_NR5GC





After Change:
	function fl_TC_8_1_4_4_4_TestBody() runs on NR5GC_PTC
  {
    var template (value) NR_CellPowerList_Type v_CellPowerList_AtT1;
    var template (value) NR_CellPowerList_Type v_CellPowerList_AtT2;
    var template (value) NR_CellPowerList_Type v_CellPowerList_AtT3;
    var NR_AbsoluteCellPower_Type v_CellPower_Serving_FR1;
    var NR_AbsoluteCellPower_Type v_CellPower_EntryCondition_FR1;
    var NR_AbsoluteCellPower_Type v_CellPower_Leaving_FR1;
    var RSRP_Range v_Rsrp;
    var template (value) MeasConfig v_MeasConfig;
    var template (value) ConditionalReconfiguration_r16 v_ConditionalReconfiguration_r16_Step;
    //var template (value) ConditionalReconfiguration_r16 v_ConditionalReconfiguration_r16_Step_5;
    var template (value) ConditionalReconfiguration_r16 v_ConditionalReconfiguration_r16_Step_10;
    var template (value) NR_DRBInfoList_Type v_DRBInfoList := f_NR5GC_MobileInfo_GetDRBInfoList();
    var template (value) DL_DCCH_Message v_RRCReconfigurationHO;
    var float v_T304Min := f_NR_SetTimerToleranceMin(nr_Cell1, rrcTimer, 1.0); //T304 is ms1000 according to 38.508-1 Table 4.6.3-19.
    var float v_T311Min := f_NR_SetTimerToleranceMin(nr_Cell1, rrcTimer, 1.0); //T311 is ms1000 according to 38.508-1 Table 4.6.3-150.
    var float v_T304_T311Sum := v_T304Min + v_T311Min;
    timer t_Watchdog := 10.0;
    timer t_T304Min;
    var integer v_DRBId;
    var PDUSessionInfoList_Type v_PDUSessionInfoList;
    var default v_MyDefaultVar := null;

    v_PDUSessionInfoList := f_NR5GC_MobileInfo_GetSessionInfoList();

    //Power levels of NR cell 1, NR cell 2 and NR cell 4 FR1 acc. Table 8.1.4.4.1.3.2-1, Table 8.1.4.4.1.3.2-2 FFS
    /*         |   Cell 1   |  Cell 2  |  Cell 4
     * -----------------------------------------
     *   T0    |    -85     |   -113   |   -88
     * -----------------------------------------
     *   T1    |    -85     |   -79    |   -113
     * -----------------------------------------
     *   T2    |    -113     |   -113   |   -113  // Change
     * -----------------------------------------
     *   T3    |    -113     |   -113   |   -79   // Change
     */

    v_CellPower_Serving_FR1 := -85;
    v_CellPower_Leaving_FR1 := -113;
    v_CellPower_EntryCondition_FR1 := -79;

    v_CellPowerList_AtT1 := {cs_NR_CellPower(nr_Cell1, v_CellPower_Serving_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE),  // FR2 is FFS
                             cs_NR_CellPower(nr_Cell2, v_CellPower_EntryCondition_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE), // FR2 is FFS
                             cs_NR_CellPower(nr_Cell4, v_CellPower_Leaving_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE)}; // FR2 is FFS
      
     v_CellPowerList_AtT2 := {cs_NR_CellPower(nr_Cell1, v_CellPower_Leaving_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE),  // FR2 is FFS
                             cs_NR_CellPower(nr_Cell2, v_CellPower_Leaving_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE), // FR2 is FFS
                             cs_NR_CellPower(nr_Cell4, v_CellPower_Leaving_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE)}; // FR2 is FFS
      
    v_CellPowerList_AtT3 := {cs_NR_CellPower(nr_Cell1, v_CellPower_Leaving_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE),  // FR2 is FFS
                             cs_NR_CellPower(nr_Cell2, v_CellPower_Leaving_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE),  // FR2 is FFS
                             cs_NR_CellPower(nr_Cell4, v_CellPower_EntryCondition_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE)}; // FR2 is FFS

    if(f_NR_CellInfo_GetIsFR1(nr_Cell1)){
       v_Rsrp := 2;
    }
    else{
       v_Rsrp := 2;          //FFS dummy value FR2
    }

    // Reconfigure nr_Cell2 to no respond to any RA Preamble
    // Table 8.1.4.4.4.3.2-4: The UE attempts to perform the inter frequency handover using MAC Random Access Preamble on NR Cell 1. SS does not respond.
    f_NR_SS_RachProcedureConfig(nr_Cell1, cs_NR_RachProcedureConfig_NoResponse);

    //@siclog "Steps 1-2" siclog@
    //The SS transmits an RRCReconfiguration message including MeasConfig and ConditionalReconfiguration to set NR Cell 2 as the target candidate cell.
    //The UE transmits an RRCReconfigurationComplete message in NR cell 1.
    v_MeasConfig := f_NR_GenerateMeasurementConfigNR( {cs_MeasObjectId1(nr_Cell1)},
                                                      {cs_NR_ReportConfigCondTrigger(tsc_NR_IdReportConfigId1, v_Rsrp)},
                                                      {cs_NR_MeasId_Config_id1_obj1_conf1}
                                                      );
    v_ConditionalReconfiguration_r16_Step := cs_ConditionalReconfiguration_r16( -,
                                                                                cs_CondReconfigToAddModList_r16_1EntryCR_1EntryMI(1,
                                                                                                                                  tsc_NR_MeasId1,
                                                                                                                                  f_NR_GenerateCondRRCReconfig_r16_Octetstring(nr_Cell2)));
    SRB.send(cas_NR_SRB1_RrcPdu_REQ( nr_Cell1,
                                     cs_TimingInfo_Now,
                                     cs_38508_RRCReconfiguration_CHO(-,v_MeasConfig, v_ConditionalReconfiguration_r16_Step)));
    SRB.receive(car_NR_SRB1_RrcPdu_IND(nr_Cell1, cr_38508_RRCReconfigurationComplete));

    //@siclog "Steps 3-4" siclog@
    //The SS transmits an RRCReconfiguration message including a reconfigurationWithSync for handover to NR Cell 4.
    //Check: Does UE transmit the RRCReconfigurationComplete message in NR Cell 4 within 10s?
    f_NR5GC_508RRC_IntraNR_HO_InterCell(nr_Cell1,
                                        nr_Cell4,
                                        -,
                                        -, // RBConfig_KeyChange := true
                                        -,
                                        -,
                                        -,
                                        cs_MasterKeyUpdate(false, tsc_NR_38508_NextHopChainingCount)
                                        );
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 4");
      
    // @siclog "Steps 4A" siclog@
    // Check: Does the test result of generic test procedure in TS 38.508-1 Table 4.9.1-1 indicate that the UE is capable of exchanging IP data on DRB1 and NR Cell 4?
    v_DRBId := f_NR_GetDefaultDRB_ForFirstPDUSession();
    f_NR5GC_CheckDataPath(nr_Cell4,
                          f_NR5GC_GetPdnIndex(v_PDUSessionInfoList[0]),
                          dedicatedBearer,
                          v_DRBId,
                          "Step 4A");

    //@siclog "Steps 5-6" siclog@
    //The SS transmits an RRCReconfiguration message including MeasConfig and ConditionalReconfiguration to set NR Cell 2 as the target candidate cell.
    //The UE transmits an RRCReconfigurationComplete message in NR Cell 4.
    //v_ConditionalReconfiguration_r16_Step_5 := cs_ConditionalReconfiguration_r16( -,
      //                                                                           cs_CondReconfigToAddModList_r16_1EntryCR_1EntryMI(1,
       //                                                                                                                             tsc_NR_MeasId1,
                                                                                                                                  //  f_NR_GenerateCondRRCReconfig_r16_Octetstring(nr_Cell2)));
    f_NR5GC_CHO_RRCReconfiguration_SSConfig( nr_Cell4,
                                             nr_Cell2,
                                             v_DRBInfoList,
                                             -,-,
                                             cs_38508_RRCReconfiguration_CHO(-,v_MeasConfig, v_ConditionalReconfiguration_r16_Step)
                                             );
      
    

    //@siclog "Steps 7" siclog@
    // Enable PRACH reception indication for NR Cell1
    f_NR_SS_SystemIndCtrlConfig(nr_Cell1, cds_NR_SystemIndCtrl_RachPreamble(enable));
     
      
    //The SS transmits an RRCReconfiguration message including a reconfigurationWithSync for handover to NR Cell 1.
    v_RRCReconfigurationHO := f_NR5GC_RRCReconfiguration_IntraNR_HO(nr_Cell1);
    SRB.send(cas_NR_SRB1_RrcPdu_REQ(nr_Cell4, cs_TimingInfo_Now, v_RRCReconfigurationHO));
    //t_T304Min.start(v_T304Min);

    //EXCEPTION: In parallel to the events described in step 10 the steps specified in Table 8.1.4.4.4.3.2-4 should take place.
    //- EXCEPTION: The steps below are repeated for the duration of T304.
     t_T304Min.start(v_T304Min);
     v_MyDefaultVar := activate ( a_NR_SysIND_Rach(nr_Cell1));
     t_T304Min.timeout;
    
     f_NR_SS_SystemIndCtrlConfig(nr_Cell1, cds_NR_SystemIndCtrl_RachPreamble(disable));
     
      deactivate(v_MyDefaultVar);
      
    //@siclog "Step 8" siclog@
    //SS adjusts the cell-specific reference signal level according to row "T1".
    f_NR_SetCellPowerList(v_CellPowerList_AtT1);

         
     
    //@siclog "Step 9" siclog@
    //Check: Does UE transmit the RRCReconfigurationComplete message in NR Cell 2 within 10s?
    t_Watchdog.start;
    f_NR5GC_CHO_RRCReconfigurationComplete( nr_Cell4,
                                            nr_Cell2,
                                            -);
    t_Watchdog.stop;
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 9");

    
    // @siclog "Steps 9A" siclog@
    // Check: Does the test result of generic test procedure in TS 38.508-1 Table 4.9.1-1 indicate that the UE is capable of exchanging IP data on DRB1 and NR Cell 2?
    //v_DRBId := f_NR_GetDefaultDRB_ForFirstPDUSession();
    f_NR5GC_CheckDataPath(nr_Cell2,
                          f_NR5GC_GetPdnIndex(v_PDUSessionInfoList[0]),
                          dedicatedBearer,
                          v_DRBId,
                          "Step 9A");
      
    //@siclog "Steps 10-11" siclog@
    //The SS transmits an RRCReconfiguration message including MeasConfig and ConditionalReconfiguration to set NR Cell 4 as the target candidate cell.
    //The UE transmits an RRCReconfigurationComplete message in NR Cell 2.
    v_ConditionalReconfiguration_r16_Step_10 := cs_ConditionalReconfiguration_r16( -,
                                                                                  cs_CondReconfigToAddModList_r16_1EntryCR_1EntryMI(1,
                                                                                                                                    tsc_NR_MeasId1,
                                                                                                                                    f_NR_GenerateCondRRCReconfig_r16_Octetstring(nr_Cell4)));
    SRB.send(cas_NR_SRB1_RrcPdu_REQ( nr_Cell2,
                                     cs_TimingInfo_Now,
                                     cs_38508_RRCReconfiguration_CHO(-,v_MeasConfig, v_ConditionalReconfiguration_r16_Step_10)));
    SRB.receive(car_NR_SRB1_RrcPdu_IND(nr_Cell2, cr_38508_RRCReconfigurationComplete));

    //@siclog "Steps 12" siclog@
     f_NR_SS_SystemIndCtrlConfig(nr_Cell1, cds_NR_SystemIndCtrl_RachPreamble(enable));
    //The SS transmits an RRCReconfiguration message including a reconfigurationWithSync for handover to NR Cell 1.
    v_RRCReconfigurationHO := f_NR5GC_RRCReconfiguration_IntraNR_HO(nr_Cell1);
    SRB.send(cas_NR_SRB1_RrcPdu_REQ(nr_Cell2, cs_TimingInfo_Now, v_RRCReconfigurationHO));

    //EXCEPTION: In parallel to the events described in step 16 the steps specified in Table 8.1.4.4.4.3.2-4 should take place.

    //@siclog "Step 13" siclog@
    
    //Note 1: The wait time is selected to cover T304 (1000ms) + T311 (1000ms) to ensure that UE goes to RRC_IDLE.
    //- EXCEPTION: The steps below are repeated for the duration of T304.
     t_T304Min.start(v_T304Min);
     v_MyDefaultVar := activate ( a_NR_SysIND_Rach(nr_Cell1));
     t_T304Min.timeout;
    
     
     f_NR_SS_SystemIndCtrlConfig(nr_Cell1, cds_NR_SystemIndCtrl_RachPreamble(disable));
     deactivate(v_MyDefaultVar);
      
     //@siclog "Step 13A" siclog@
    //SS adjusts the cell-specific reference signal level according to row "T2".
     f_NR_SetCellPowerList(v_CellPowerList_AtT2); // No selectable cell available to UE now
      
     f_NR_RRCRelease_Local(nr_Cell4, f_NR_GetNextSendOccasion_DL(nr_Cell4));
     
      //SS wait 5s to make sure T304+T311 expire. (Note 1)
     f_Delay(5.0); 

    //@siclog "Step 14" siclog@
    //SS adjusts the cell-specific reference signal level according to row "T3".
    f_NR_SetCellPowerList(v_CellPowerList_AtT3);
 
	//@siclog "Step 15" siclog@
    //Check: Does the generic mobility registration updating procedure described in TS 38.508-1 [4] Table 4.9.5.2.2-1 is performed on NR Cell 4.
    f_NR5GC_MobilityRegistration(nr_Cell4);
  } // End of f_TC_8_1_4_4_4_NR5GC

}



2.2 [bookmark: _Toc97128121][bookmark: _Toc97906427]Implemented new altstep a_NR_SysIND_Rach ()
	Function name
	altstep a_NR_SysIND_Rach

	Reason for change
	Altstep is needed to cater the repeated need of Step 7 and Step 12 for the repeated duration of T304 timer.


	Summary of change
	Added an altstep to receive Rach preamble.



	TTCN module
	RRC_Conditional_Handover_NR5GC.ttcn

	MCC160 Comment
	1. Accepted in principle, considering function fl_NR_SysIND_Rach has been used in TC 8.1.4.3.3, we’d better to reuse fl_NR_SysIND_Rach here.



    
New implementation
	    // WA#WI=933863
  altstep a_NR_SysIND_Rach(NR_CellId_Type p_NR_CellId) runs on NR_BASE_PTC
  {
    [] SYSIND.receive(car_NR_PRACH_Preamble_IND(p_NR_CellId, cr_TimingInfo_Any, ?))
      {

        repeat;
      }
  };




1. [bookmark: _Toc54888065][bookmark: _Toc97717938][bookmark: _Toc97906428][bookmark: _Toc122434494][bookmark: _Toc295288971][bookmark: _Toc325725666]Branches executed
This NR5GC TC was executed in 5GS Stand Alone mode (Option 2) on NR Primary Band “n78” using NIA2 Integrity and NEA2 ciphering algorithms
4 [bookmark: _Toc97717939][bookmark: _Toc97906429]Execution Log Files 
4.1 [bookmark: _Toc97717940][bookmark: _Toc97906430]MediaTek MT6983
The MediaTek MT6983 UE passed this test case on R&S® 5G Protocol Conformance Test platform. The documentation below is enclosed as evidence of the successful test case run [1]:
Test case execution log file:
TC_8_1_4_4_4_NR5GC_MTK.log
(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)
In the log file the complete test case execution as well as the PICS/PIXIT parameter settings can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.
The MediaTek MT6983 UE passed this test case on Anritsu Protocol Conformance Test System ME7834NR solution. The documentation below is enclosed as evidence of the successful test case run [1]:
· Test case execution log file:
Anritsu/TC_8_1_1_4_4_4_LOG.xml
· PICS/PIXIT parameter file:
Anritsu/TC_8_1_1_4_4_4_PICS.txt

5 [bookmark: _Toc122434496][bookmark: _Toc295288973][bookmark: _Toc325725668][bookmark: _Toc97717941][bookmark: _Toc97906431]References

